

Programming Paradigms
DUE SEPTEMBER 21

The Game of Twenty-one

NOTE: the code for this assignment is in \$CLASSHOME/hw3/hw3.scm. Please make a copy to work on in your own directory first. At this point, you should also have set your language level in DrScheme to Full Scheme, Graphical (MrEd).

Louis Reasoner took a course on game theory and became interested in the card game Twenty-One (also called Blackjack). Louis was also treasurer of his fraternity. By the end of the semester, he had managed to squander the term's dinner money at Atlantic City casinos in an attempt to perfect his "no-lose" strategy. Ben Bitdiddle, Louis's roommate, has decided to construct a general-purpose Twenty-One simulator to help discover what Louis has been doing wrong (in terms of playing Twenty-One, not in terms of moral or ethical conduct).

For our purposes, the rules of Twenty-One are as follows. There are two players, and the object of the game is to be dealt a set of cards that totals as close to 21 as possible without going over 21, where each number card has that number as its value, and face cards have a value of 10. Each player is dealt one card face up that the other player can see. Subsequent cards are dealt face down. One player plays first, asking for more cards one at a time (called a "hit") until he decides to "stay" with the total he has or until his total exceeds 21. If a player's total exceeds 21 he "busts", meaning he immediately loses the game. If the first player does not bust, the second player, called the *house*, then plays, asking for more cards until either losing by exceeding 21 or deciding to stay with the current total. After the house decides not to take additional cards both players expose their cards, and the player with the largest total wins. In the event of a tie, the house wins. This version of Twenty-One is simplified: we will not consider such things as "splitting" or the special treatment of aces. In fact, since we are interested only in the relative strengths of competing strategies, we will not simulate betting either.

A player's *strategy* determines when he wishes another card and when he would like to stay with what he has. For our Scheme simulation, each player will be modeled by a procedure that implements his strategy. Since a typical strategy for when to stay and when to hit involves both the player's current hand and the point value of the opponent's face-up card, we will represent a strategy as a procedure of two arguments: the player's hand and the point value of the opponent's face-up card. The procedure returns true if the player would want another card, and false if the player would stay. For example, the following (stupid) strategy procedure will always take a card if the opponent's up card is greater than 5:

```
(define stupid-strategy
  (lambda (my-hand opponent-up-card)
    (> opponent-up-card 5)))
```

The following procedure `play-hand` takes as arguments a strategy, a hand, and the opponent's up card. It continues to accept cards for as long as the strategy requests, or until the total of the cards in the hand exceeds 21. `play-hand` returns as a value, the full hand that was dealt.

```
(define play-hand
  (lambda (strategy my-hand opponent-up-card)
    (cond ((> (hand-total my-hand) 21) my-hand) ; I lose... give up
          ((strategy my-hand opponent-up-card) ; hit?
           (play-hand strategy
                       (hand-add-card my-hand (deal))
                       opponent-up-card))
          (else my-hand)))) ; stay
```

For the purposes of this simple simulation, a “hand” of cards will be represented by two numbers—the value of the up card and the total of all the cards in the hand. We will represent this using a very simple form of *data abstraction*, of the kind that will be discussed in lecture today—we have a *constructor* procedure `make-hand` that creates a hand from two numbers, and two *selectors* `hand-up-card` and `hand-total` that return the up card and total of a given hand:

```
(define make-hand
  (lambda (up-card total)
    (cons up-card total)))

(define hand-up-card
  (lambda (hand)
    (car hand)))

(define hand-total
  (lambda (hand)
    (cdr hand)))
```

Don’t worry right now about the details of the Scheme primitives `cons`, `car`, and `cdr`, which are used in implementing these procedures—we will be discussing this topic in all too much detail over the next few weeks. For now, consider `make-hand`, `hand-up-card`, and `hand-total` to be simple “black boxes” that allow us to represent hands.

In terms of these basic procedures, we can implement some useful operations on hands. `Make-new-hand` takes as argument a first card and returns a hand containing only that card (i.e., the card is both the up card and the total):

```
(define make-new-hand
  (lambda (first-card)
    (make-hand first-card first-card)))
```

`Hand-add-card` takes a hand and a new card and returns a hand with the same up-card as the original, but with the total augmented by the value of the new card:

```
(define hand-add-card
  (lambda (hand new-card)
    (make-hand (hand-up-card hand)
               (+ new-card (hand-total hand)))))
```

Instead of modeling a real deck of cards, we simply deal cards at random from an infinite deck in which each card value from 1 to 10 is equally probable. (This does not, of course, correctly model real decks of cards, but since our focus is one of strategies, we won’t worry about the difference.) We represent dealing a card as simply returning a random number in the range 1 through 10:

```
(define deal
  (lambda ()
    (+ 1 (random 10))))
```

Finally, the top-level procedure in our simulation, `twenty-one`, simulates one game of Twenty-One. It takes strategy procedures for a player and for the house as its two arguments. It creates initial hands for the house and the player, then plays the player strategy, then plays the house strategy. `Twenty-one` returns 1 if the player wins the simulated game and 0 if the house wins.

```
(define twenty-one
  (lambda (player-strategy house-strategy)
    (let ((house-initial-hand (make-new-hand (deal)))) ; set up house hand
      ; let is covered on pp.58-61 of text
      (let ((player-hand ; set up initial hand, and play out
              (play-hand player-strategy ; strategy to use
                          (make-new-hand (deal)) ; initial player hand
                          (hand-up-card house-initial-hand)))) ;
          ;information about house hand available to player
          (if (> (hand-total player-hand) 21)
              0 ; ``bust``: player loses
              (let ((house-hand ; play out house hand
                      (play-hand house-strategy
                                  house-initial-hand
                                  (hand-up-card player-hand))))
                  (cond ((> (hand-total house-hand) 21)
                         1) ; ``bust``: house loses
                        ((> (hand-total player-hand)
                            (hand-total house-hand))
                         1) ; house loses
                        (else 0))))))) ; player loses
```

`Hit?` is a simple interactive strategy procedure that can be used with `twenty-one`. It displays on the screen the information available to the player it is simulating and asks in a dialog box whether it should take another card. It returns true if you press the “Hit Me” button and false if you press the “Stay” button.¹ Remember that the last combination of a `define` or `let` with multiple combinations is the value that is returned by the procedure or `let`. There are two versions of the `user-says-y` function, depending on if you are using Dr.Scheme V52 or V100 or higher!! Everyone should be using DrScheme v102 or v103 at this time.

```
(define hit?
  (lambda (your-hand opponent-up-card)
    (newline)
    (display "Opponent up card ")
    (display opponent-up-card)
    (newline)
    (display "Your Total: ")
    (display (hand-total your-hand))
    (user-says-y?)))
```

¹Sometimes the dialog box will pop up before the displayed text is all printed out! Be patient!

Problem 1

Load in the code for problem set 2 and try playing a few games of Twenty-One against yourself by evaluating:

```
(twenty-one hit? hit?)
```

Remember that the first set of questions you will be asked are for the player's hand and the second set of questions are for the house's hand. There is nothing to turn in for this problem.

Problem 1.5

Define a twenty-one game *strategy* procedure `stop-at-16` that takes two arguments (like every strategy procedure does). The first argument is a hand, and the second argument is the opponent-up-card (see the example `stupid-strategy` earlier in the handout). Your new strategy procedure should ask for a new card if and only if the total of the hand is less than 16. You can play a few games by evaluating

```
(twenty-one hit? stop-at-16)
```

Please note the difference between Problem 1.5 and the following Problem 2!!!! Turn in a listing of your procedure.

Problem 2

Define a procedure `stop-at` that takes a number as its single argument and returns a strategy procedure. The strategy procedure created by `stop-at` should ask for a new card if and only if the total of a hand is less than the argument to `stop-at`. For example `(stop-at 16)` should return a strategy that asks for another card if the hand total is less than 16, but stops as soon as the total reaches 16. To test your implementation of `stop-at`, play a few games by evaluating

```
(twenty-one hit? (stop-at 16))
```

Thus, you will be playing against a house whose strategy is to stop at 16. Turn in a listing of your procedure.

Problem 3

Define a procedure `test-strategy` that tests two strategies by playing a specified number of simulated Twenty-One games using the two strategies. `Test-strategy` should return the number of games that were won by the player (and thus lost by the house). For example,

```
(test-strategy (stop-at 16) (stop-at 15) 10)
```

should play ten games of Twenty-One, using the value returned by `(stop-at 16)` as the player's strategy and the value of `(stop-at 15)` as the house strategy. It should return a non-negative integer indicating how many games were won by the player. Turn in a listing of your procedure and some sample results.

Problem 4

When the simulated games in the previous Problem ran, it was impossible for us to tell what was going on. It would be nice if we could watch a strategy play by observing its inputs and the decisions it makes. Define a procedure called `watch-player` that takes a strategy as an argument and returns a strategy as its result. The strategy returned by `watch-player` should implement the same result as the strategy that was passed to it as an argument, but, in addition, it should print the information supplied to the strategy and the decision that the strategy returns. For example,

```
(test-strategy (watch-player (stop-at 16))
               (watch-player (stop-at 15))
               2)
```

should play two simulated games and show what each player does at each step. Turn in a listing of your procedure and some sample runs using it.

Problem 5

Ben has finally gotten Louis to describe his Twenty-One strategy. Here is how Louis was playing. If his hand contained fewer than 12 points, he always asked for another card. If his hand had more than 16 points, he always stayed with what he had. If his hand had exactly 12 points, he took another card if his opponent's up card was less than 4. If his hand had exactly 16 points, he would stay if his opponent was showing 10. If none of the above conditions held (his hand had between 12 and 16 points, exclusive), Louis would take a card if his opponent's up card was greater than 6, otherwise he would stay with what he had. Define a procedure called `louis` that implements Louis's strategy. Try Louis's strategy against the strategies of stopping at 15, 16, and 17 by evaluating

```
(test-strategy louis (stop-at 15) 10)
(test-strategy louis (stop-at 16) 10)
(test-strategy louis (stop-at 17) 10)
```

Problem 6

Implement a procedure `both` that takes two strategies as arguments and returns a new strategy. This new strategy will call for a new card if and only if *both* strategies would ask for a new card. For example, using the strategy

```
(both (stop-at 19) hit?)
```

will ask for a new card only if the hand total is less than 19 and the user requests a hit. Turn in a listing of your procedure and an example showing that it works.